Rules for integrands of the form $P[x]\ (d+e\,x)^q\ \left(a+b\,x^2+c\,x^4\right)^p$

1. $\int (d+e\,x)^q\ (a+b\,x^2+c\,x^4)^p\, dx$

   1. $\int \frac{(d+e\,x)^q}{\sqrt{a+b\,x^2+c\,x^4}}\, dx$

      1: $\int \frac{1}{(d+e\,x)\ \sqrt{a+b\,x^2+c\,x^4}}\, dx$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e\,x} == \frac{d}{d^2-e^2\,x^2} - \frac{e\,x}{d^2-e^2\,x^2}$

Rule 1.2.2.5.2.1:

$$\int \frac{1}{(d+e\,x)\ \sqrt{a+b\,x^2+c\,x^4}}\, dx \longrightarrow d\int \frac{1}{\left(d^2-e^2\,x^2\right)\ \sqrt{a+b\,x^2+c\,x^4}}\, dx - e\int \frac{x}{\left(d^2-e^2\,x^2\right)\ \sqrt{a+b\,x^2+c\,x^4}}\, dx$$

Program code:

```
Int[1/((d_+e_.*x_)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  d*Int[1/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x] - e*Int[x/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e},x]
```

```
Int[1/((d_+e_.*x_)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  d*Int[1/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x] - e*Int[x/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e},x]
```

      2: $\int \frac{(d+e\,x)^q}{\sqrt{a+b\,x^2+c\,x^4}}\, dx$ when $c\,d^4+b\,d^2\,e^2+a\,e^4\neq 0\ \land\ q<-1$

Derivation: Algebraic expansion

Rule 1.2.2.5.2.2.2: If $c\,d^4+b\,d^2\,e^2+a\,e^4\neq 0\ \land\ q<-1$, then

$$\int \frac{(d+e\,x)^q}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \longrightarrow$$

$$\frac{e^3\,(d+e\,x)^{q+1}\,\sqrt{a+b\,x^2+c\,x^4}}{(q+1)\,\left(c\,d^4+b\,d^2\,e^2+a\,e^4\right)}+$$

$$\frac{1}{(q+1)\,\left(c\,d^4+b\,d^2\,e^2+a\,e^4\right)}\int \frac{(d+e\,x)^{q+1}}{\sqrt{a+b\,x^2+c\,x^4}}\left(d\,(q+1)\,\left(c\,d^2+b\,e^2\right)-e\,\left(c\,d^2\,(q+1)+b\,e^2\,(q+2)\right)\,x+c\,d\,e^2\,(q+1)\,x^2-c\,e^3\,(q+3)\,x^3\right)dx$$

–

Program code:

```
Int[(d_+e_.*x_)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  e^3*(d+e*x)^(q+1)*Sqrt[a+b*x^2+c*x^4]/((q+1)*(c*d^4+b*d^2*e^2+a*e^4)) +
  1/((q+1)*(c*d^4+b*d^2*e^2+a*e^4))*
    Int[(d+e*x)^(q+1)/Sqrt[a+b*x^2+c*x^4]*
      Simp[d*(q+1)*(c*d^2+b*e^2)-e*(c*d^2*(q+1)+b*e^2*(q+2))*x+c*d*e^2*(q+1)*x^2-c*e^3*(q+3)*x^3,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c*d^4+b*d^2*e^2+a*e^4,0] && ILtQ[q,-1]
```

```
Int[(d_+e_.*x_)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  e^3*(d+e*x)^(q+1)*Sqrt[a+c*x^4]/((q+1)*(c*d^4+a*e^4)) +
  c/((q+1)*(c*d^4+a*e^4))*
    Int[(d+e*x)^(q+1)/Sqrt[a+c*x^4]*Simp[d^3*(q+1)-d^2*e*(q+1)*x+d*e^2*(q+1)*x^2-e^3*(q+3)*x^3,x],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^4+a*e^4,0] && ILtQ[q,-1]
```

2: $\int \frac{\left(a+b x^2+c x^4\right)^p}{d+e x}\,dx$ when $p+\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x}==\frac{d}{d^2-e^2 x^2}-\frac{e x}{d^2-e^2 x^2}$

Rule 1.2.2.5.1: If $p+\frac{1}{2}\in\mathbb{Z}$, then

$$\int \frac{\left(a+b x^2+c x^4\right)^p}{d+e x}\,dx \longrightarrow d\int \frac{\left(a+b x^2+c x^4\right)^p}{d^2-e^2 x^2}\,dx-e\int \frac{x\left(a+b x^2+c x^4\right)^p}{d^2-e^2 x^2}\,dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_./(d_+e_.*x_),x_Symbol] :=
  d*Int[(a+b*x^2+c*x^4)^p/(d^2-e^2*x^2),x] - e*Int[x*(a+b*x^2+c*x^4)^p/(d^2-e^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p+1/2]
```

```
Int[(a_+c_.*x_^4)^p_./(d_+e_.*x_),x_Symbol] :=
  d*Int[(a+c*x^4)^p/(d^2-e^2*x^2),x] - e*Int[x*(a+c*x^4)^p/(d^2-e^2*x^2),x] /;
FreeQ[{a,c,d,e},x] && IntegerQ[p+1/2]
```

2: $\int P[x]\,(d+e\,x)^q\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$ when PolynomialRemainder[P[x], d + e x, x] == 0

Derivation: Algebraic simplification

Rule: If PolynomialRemainder[P[x], d + e x, x] == 0, then

$$\int P[x]\,(d+e\,x)^q\,\left(a+b\,x^2+c\,x^4\right)^p\,dx\;\longrightarrow\;\int \text{PolynomialQuotient}[P[x],\,d+e\,x,\,x]\,(d+e\,x)^{q+1}\,\left(a+b\,x^2+c\,x^4\right)^p\,dx$$

Program code:

```
Int[Px_*(d_+e_.*x_)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,d+e*x,x]*(d+e*x)^(q+1)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,d+e*x,x],0]
```

```
Int[Px_*(d_+e_.*x_)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,d+e*x,x]*(d+e*x)^(q+1)*(a+c*x^4)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,d+e*x,x],0]
```

3: $\int P[x]\,(d+e\,x)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when PolynomialRemainder$[P[x],\ a+b\,x^2+c\,x^4,\ x] == 0$

- Derivation: Algebraic simplification

- Rule: If PolynomialRemainder$[P[x],\ a+b\,x^2+c\,x^4,\ x] == 0$, then

$$\int P[x]\,(d+e\,x)^q\,(a+b\,x^2+c\,x^4)^p\,dx \rightarrow \int \text{PolynomialQuotient}[P[x],\ a+b\,x^2+c\,x^4,\ x]\,(d+e\,x)^q\,(a+b\,x^2+c\,x^4)^{p+1}\,dx$$

- Program code:

```
Int[Px_*(d_+e_.*x_)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x^2+c*x^4,x]*(d+e*x)^q*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x^2+c*x^4,x],0]
```

```
Int[Px_*(d_+e_.*x_)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+c*x^4,x]*(d+e*x)^q*(a+c*x^4)^(p+1),x] /;
FreeQ[{a,c,d,e,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+c*x^4,x],0]
```

4. $\int \frac{(d+e\,x)^q\,(A+B\,x+C\,x^2+D\,x^3)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $c\,d^4+b\,d^2\,e^2+a\,e^4 \neq 0$

1. $\int \frac{(d+e\,x)^q\,(A+B\,x+C\,x^2+D\,x^3)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $c\,d^4+b\,d^2\,e^2+a\,e^4 \neq 0 \land q>0$

1: $\int \frac{(d+e\,x)^q\,(A+B\,x+C\,x^2)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $c\,d^4+b\,d^2\,e^2+a\,e^4 \neq 0 \land q>0$

- Derivation: Algebraic expansion

- Basis: $(d+e\,x)\,(A+B\,x+C\,x^2) == A\,d+(B\,d+A\,e)\,x+(C\,d+B\,e)\,x^2+C\,e\,x^3$

- Rule 1.2.2.5.2.2.2: If $c\,d^4+b\,d^2\,e^2+a\,e^4 \neq 0 \land q>0$, then

$$\int \frac{(d+e\,x)^q\,\left(A+B\,x+C\,x^2\right)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \longrightarrow \int \frac{(d+e\,x)^{q-1}\,\left(A\,d+\,(B\,d+A\,e)\,x+\,(C\,d+B\,e)\,x^2+C\,e\,x^3\right)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$$

–

Program code:

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  Int[(d+e*x)^(q-1)*(A*d+(B*d+A*e)*x+(C*d+B*e)*x^2+C*e*x^3)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],2] && NeQ[c*d^4+b*d^2*e^2+a*e^4,0] && GtQ[q,0]
```

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  Int[(d+e*x)^(q-1)*(A*d+(B*d+A*e)*x+(C*d+B*e)*x^2+C*e*x^3)/Sqrt[a+c*x^4],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],2] && NeQ[c*d^4+a*e^4,0] && GtQ[q,0]
```

2: $\int \frac{(d+e x)^q \left(A+B x+C x^2+D x^3\right)}{\sqrt{a+b x^2+c x^4}} dx$ when $c d^4+b d^2 e^2+a e^4 \neq 0 \wedge q>0$

Rule 1.2.2.5.2.2.2: If $c d^4+b d^2 e^2+a e^4 \neq 0 \wedge q>0$, then

$$\int \frac{(d+e x)^q \left(A+B x+C x^2+D x^3\right)}{\sqrt{a+b x^2+c x^4}} dx \longrightarrow$$

$$\frac{D (d+e x)^q \sqrt{a+b x^2+c x^4}}{c (q+2)}-\frac{1}{c (q+2)} \int \frac{(d+e x)^{q-1}}{\sqrt{a+b x^2+c x^4}} \cdot$$

$$\left(a D e q-A c d (q+2)+(b d D-B c d (q+2)-A c e (q+2)) x+(b D e (q+1)-c (C d+B e) (q+2)) x^2-c (d D q+C e (q+2)) x^3\right) dx$$

Program code:

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  D*(d+e*x)^q*Sqrt[a+b*x^2+c*x^4]/(c*(q+2)) -
  1/(c*(q+2))*Int[(d+e*x)^(q-1)/Sqrt[a+b*x^2+c*x^4]*
    Simp[a*D*e*q-A*c*d*(q+2)+(b*d*D-B*c*d*(q+2)-A*c*e*(q+2))*x+
      (b*D*e*(q+1)-c*(C*d+B*e)*(q+2))*x^2-c*(d*D*q+C*e*(q+2))*x^3,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x,3] && NeQ[c*d^4+b*d^2*e^2+a*e^4,0] && GtQ[q,0]
```

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  D*(d+e*x)^q*Sqrt[a+c*x^4]/(c*(q+2)) -
  1/(c*(q+2))*Int[(d+e*x)^(q-1)/Sqrt[a+c*x^4]*
    Simp[a*D*e*q-A*c*d*(q+2)-c*(B*d*(q+2)+A*e*(q+2))*x-c*(C*d+B*e)*(q+2)*x^2-c*(d*D*q+C*e*(q+2))*x^3,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x,3] && NeQ[c*d^4+a*e^4,0] && GtQ[q,0]
```

2: $\int \frac{(d+e\,x)^q\,\left(A+B\,x+C\,x^2+D\,x^3\right)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $c\,d^4+b\,d^2\,e^2+a\,e^4\neq 0\,\wedge\,q<-1$

- Note: If $d^3\,D-C\,d^2\,e+B\,d\,e^2-A\,e^3=0$, then PolynomialRemainder$\left[A+B\,x+C\,x^2+D\,x^3,\,d+e\,x,\,x\right]=0$.

- Rule 1.2.2.5.2.2.2: If $c\,d^4+b\,d^2\,e^2+a\,e^4\neq 0\,\wedge\,q<-1$, then

$$\int \frac{(d+e\,x)^q\,\left(A+B\,x+C\,x^2+D\,x^3\right)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\,\longrightarrow$$

$$-\frac{\left(d^3\,D-C\,d^2\,e+B\,d\,e^2-A\,e^3\right)\,(d+e\,x)^{q+1}\,\sqrt{a+b\,x^2+c\,x^4}}{(q+1)\,\left(c\,d^4+b\,d^2\,e^2+a\,e^4\right)}+\frac{1}{(q+1)\,\left(c\,d^4+b\,d^2\,e^2+a\,e^4\right)}\int \frac{(d+e\,x)^{q+1}}{\sqrt{a+b\,x^2+c\,x^4}}\,\cdot$$
$$\Big((q+1)\,\left(a\,e\,\left(d^2\,D-C\,d\,e+B\,e^2\right)+A\,d\,\left(c\,d^2+b\,e^2\right)\right)-$$
$$\left(e\,(q+1)\,\left(A\,c\,d^2+a\,e\,(d\,D-C\,e)\right)-B\,d\,\left(c\,d^2\,(q+1)+b\,e^2\,(q+2)\right)-b\,\left(d^3\,D-C\,d^2\,e-A\,e^3\,(q+2)\right)\right)\,x+$$
$$(q+1)\,\left(D\,e\,\left(b\,d^2+a\,e^2\right)+c\,d\,\left(C\,d^2-e\,(B\,d-A\,e)\right)\right)\,x^2+$$
$$c\,(q+3)\,\left(d^3\,D-C\,d^2\,e+B\,d\,e^2-A\,e^3\right)\,x^3\Big)\,dx$$

- Program code:

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  -(d^3*D-C*d^2*e+B*d*e^2-A*e^3)*(d+e*x)^(q+1)*Sqrt[a+b*x^2+c*x^4]/((q+1)*(c*d^4+b*d^2*e^2+a*e^4)) +
  1/((q+1)*(c*d^4+b*d^2*e^2+a*e^4))*
    Int[((d+e*x)^(q+1)/Sqrt[a+b*x^2+c*x^4])*
      Simp[(q+1)*(a*e*(d^2*D-C*d*e+B*e^2)+A*d*(c*d^2+b*e^2)) -
        (e*(q+1)*(A*c*d^2+a*e*(d*D-C*e))-B*d*(c*d^2*(q+1)+b*e^2*(q+2))-b*(d^3*D-C*d^2*e-A*e^3*(q+2)))*x +
        (q+1)*(D*e*(b*d^2+a*e^2)+c*d*(C*d^2-e*(B*d-A*e)))*x^2 +
        c*(q+3)*(d^3*D-C*d^2*e+B*d*e^2-A*e^3)*x^3,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],3] && NeQ[c*d^4+b*d^2*e^2+a*e^4,0] && LtQ[q,-1]
```

```
Int[Px_*(d_+e_.*x_)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  -(d^3*D-C*d^2*e+B*d*e^2-A*e^3)*(d+e*x)^(q+1)*Sqrt[a+c*x^4]/((q+1)*(c*d^4+a*e^4)) +
   1/((q+1)*(c*d^4+a*e^4))*
     Int[((d+e*x)^(q+1)/Sqrt[a+c*x^4])*
       Simp[(q+1)*(a*e*(d^2*D-C*d*e+B*e^2)+A*d*(c*d^2)) -
         (e*(q+1)*(A*c*d^2+a*e*(d*D-C*e))-B*d*(c*d^2*(q+1)))*x +
         (q+1)*(D*e*(a*e^2)+c*d*(C*d^2-e*(B*d-A*e)))*x^2 +
        c*(q+3)*(d^3*D-C*d^2*e+B*d*e^2-A*e^3)*x^3,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],3] && NeQ[c*d^4+a*e^4,0] && LtQ[q,-1]
```

3. $\int \frac{A+B x+C x^2+D x^3}{(d+e x)\sqrt{a+b x^2+c x^4}}\,dx$ when $c d^4+b d^2 e^2+a e^4\neq 0$

1: $\int \frac{A+B x}{(d+e x)\sqrt{a+b x^2+c x^4}}\,dx$ when $B d-A e\neq 0\land c^2 d^6+a e^4\left(13 c d^2+b e^2\right)==0\land b^2 e^4-12 c d^2\left(c d^2-b e^2\right)==0\land 4 A c d e+B\left(2 c d^2-b e^2\right)==0$

Derivation: Integration by substitution

Basis: If $c^2 d^6+a e^4\left(13 c d^2+b e^2\right)==0\land b^2 e^4-12 c d^2\left(c d^2-b e^2\right)==0\land 4 A c d e+B\left(2 c d^2-b e^2\right)==0$, then

$$\frac{A+B x}{(d+e x)\sqrt{a+b x^2+c x^4}}==-\frac{A^2 (B d+A e)}{e}\,\text{Subst}\left[\frac{1}{6 A^3 B d+3 A^4 e-a e x^2},\,x,\,\frac{(A+B x)^2}{\sqrt{a+b x^2+c x^4}}\right]\partial_x\frac{(A+B x)^2}{\sqrt{a+b x^2+c x^4}}$$

Rule 1.2.2.9.2.1: If $B d-A e\neq 0\land c^2 d^6+a e^4\left(13 c d^2+b e^2\right)==0\land b^2 e^4-12 c d^2\left(c d^2-b e^2\right)==0\land 4 A c d e+B\left(2 c d^2-b e^2\right)==0$, then

$$\int \frac{A+B x}{(d+e x)\sqrt{a+b x^2+c x^4}}\,dx\to-\frac{A^2 (B d+A e)}{e}\,\text{Subst}\left[\int\frac{1}{6 A^3 B d+3 A^4 e-a e x^2}\,dx,\,x,\,\frac{(A+B x)^2}{\sqrt{a+b x^2+c x^4}}\right]$$

Program code:

```
Int[(A_+B_.*x_)/((d_+e_.*x_)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  -A^2*(B*d+A*e)/e*Subst[Int[1/(6*A^3*B*d+3*A^4*e-a*e*x^2),x],x,(A+B*x)^2/Sqrt[a+b*x^2+c*x^4]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[B*d-A*e,0] && EqQ[c^2*d^6+a*e^4*(13*c*d^2+b*e^2),0] &&
  EqQ[b^2*e^4-12*c*d^2*(c*d^2-b*e^2),0] && EqQ[4*A*c*d*e+B*(2*c*d^2-b*e^2),0]
```

2: $\int \frac{A+B x+C x^2+D x^3}{(d+e x)\sqrt{a+b x^2+c x^4}}\,dx$ when $c d^4+b d^2 e^2+a e^4\neq 0$

Derivation: Algebraic expansion

- Basis: $\frac{A+B x+C x^2+D x^3}{d+e x}==\frac{x\left(B d-A e+(d D-C e)\,x^2\right)}{d^2-e^2 x^2}+\frac{A d+(C d-B e)\,x^2-D e x^4}{d^2-e^2 x^2}$

- Rule 1.2.2.5.2.2.2: If $c d^4+b d^2 e^2+a e^4\neq 0$, then

$$\int \frac{A+B x+C x^2+D x^3}{(d+e x)\sqrt{a+b x^2+c x^4}}\,dx \to \int \frac{x\left(B d-A e+(d D-C e)\,x^2\right)}{\left(d^2-e^2 x^2\right)\sqrt{a+b x^2+c x^4}}\,dx+\int \frac{A d+(C d-B e)\,x^2-D e x^4}{\left(d^2-e^2 x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$$

- Program code:

```
Int[Px_/((d_+e_.*x_)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  Int[(x*(B*d-A*e+(d*D-C*e)*x^2))/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x] +
  Int[(A*d+(C*d-B*e)*x^2-D*e*x^4)/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],3] && NeQ[c*d^4+b*d^2*e^2+a*e^4,0]
```

```
Int[Px_/((d_+e_.*x_)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2],D=Coeff[Px,x,3]},
  Int[(x*(B*d-A*e+(d*D-C*e)*x^2))/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x] +
  Int[(A*d+(C*d-B*e)*x^2-D*e*x^4)/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x] && LeQ[Expon[Px,x],3] && NeQ[c*d^4+a*e^4,0]
```

5: $\int \frac{P[x] \left(a+b x^2+c x^4\right)^p}{d+e x} \, dx$ when $p+\frac{1}{2} \in \mathbb{Z}$

– Derivation: Algebraic expansion

– Basis: $\frac{1}{d+e x} = \frac{d}{d^2-e^2 x^2} - \frac{e x}{d^2-e^2 x^2}$

– Rule 1.2.2.9.1: If $p+\frac{1}{2} \in \mathbb{Z}$, then

$$\int \frac{P[x] \left(a+b x^2+c x^4\right)^p}{d+e x} \, dx \longrightarrow d \int \frac{P[x] \left(a+b x^2+c x^4\right)^p}{d^2-e^2 x^2} \, dx - e \int \frac{x P[x] \left(a+b x^2+c x^4\right)^p}{d^2-e^2 x^2} \, dx$$

– Program code:

```
Int[Px_*(a_+b_.*x_^2+c_.*x_^4)^p_./(d_+e_.*x_),x_Symbol] :=
  d*Int[Px*(a+b*x^2+c*x^4)^p/(d^2-e^2*x^2),x] - e*Int[x*Px*(a+b*x^2+c*x^4)^p/(d^2-e^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && IntegerQ[p+1/2]
```

```
Int[Px_*(a_+c_.*x_^4)^p_./(d_+e_.*x_),x_Symbol] :=
  d*Int[Px*(a+c*x^4)^p/(d^2-e^2*x^2),x] - e*Int[x*Px*(a+c*x^4)^p/(d^2-e^2*x^2),x] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x] && IntegerQ[p+1/2]
```